Breve guida a Gnuplot a cura di Egidio Rezzaghi

GNUPLOT

GNUPLOT è un programma interattivo per il tracciamento di funzioni e la rappresentazione grafica di dati che si basa su comandi introdotti da tastiera ed interpretati istantaneamente, l'uso e la diffusione del programma sono liberi. Coi presenti appunti vengono fornite le notizie indispensabili per un uso semplice ed immediatamente produttivo di Gnuplot, chi volesse approfondirne la conoscenza può farlo attraverso la funzione di help interna al programma o attraverso la consultazione del manuale GNUPLOT.PDF che accompagna il programma stesso e che si trova assieme ad altra documentazione nella cartella C:\gnuplot\docs\. 

Come avviare GNUPLOT
Una volta spostatisi nella cartella contenente i files eseguibili del programma, di solito si tratta di C:\gnuplot\binaries\, cliccare sul file:

wgnuplot.exe

si aprirà così la finestra di editing e si entrerà nell'ambiente interattivo di Gnuplot caratterizzato carattere di prompt:

gnuplot>

D'ora in avanti si dovranno usare per i comandi i caratteri minuscoli della tastiera, quindi se è attivo il tasto  CAPS , esso dovrà essere disattivato.

Configurazione delle finestre di editing e grafica ed altre importanti opzioni
Sia la finestra di editing che la finestra grafica (quest'ultima si apre appena si invia il comando plot ed in essa si visualizza il grafico) presentano in alto a sinistra nella barra di intestazione un'icona cliccando sulla quale si apre un menù testuale che presenta tra gli altri il comando Options. Attraverso tale comando è possibile, tra le altre cose, copiare negli appunti di windows il contenuto della finestra, configurare a piacimento la visualizzazione negli ambienti di editing e grafico, stampare su carta quanto visualizzato nella finestra grafica. Se si clicca su Update wgnuplot.ini le modifiche apportate ai due ambienti vengono salvate per essere rese operative nelle successive sessioni di gnuplot.
Volendo è possibile scrivere tutti i comandi desiderati in un file testo (usando ad esempio un editor quale notepad.exe o edit.com), salvare il file testo ed infine eseguire i comandi caricando il file nell'ambiente di editing di gnuplot attraverso il comando File Open. Parimenti è possibile salvare un'intera sessione di gnuplot in un file di testo col comando File Save e rieseguirla successivamente richiamandola col comando File Open.

Come richiamare linee di istruzioni già eseguite
Durante una sessione di gnuplot, se si vogliono rieseguire comandi introdotti in una precedente linea di istruzioni basta richiamare la linea interessata mediante la pressione ripetuta una o più volte dei tasti direzionali (quelli con le freccette su e giù) e premere il tasto di invio appena compare la linea desiderata (che può essere modificata prima di venire mandata in esecuzione).

Come tracciare il grafico di una funzione del tipo  y = f(x)
Si tratta di una funzione non parametrica, ovvero di una funzione nella quale la variabile  y  è esplicitabile rispetto ad  x  e per ciascun valore di  x  si ha un solo valore di  y.

gnuplot> unset parametric
gnuplot> set grid
gnuplot> set time
gnuplot> set title "es. funzioni goniometriche"
gnuplot> set xlabel "radianti"
gnuplot> set ylabel "valori funzione"
gnuplot> set samples 200
gnuplot> set xrange [0:2*pi]
gnuplot> set yrange [-5:5]
gnuplot> plot sin(x),cos(x),tan(x)
Si osserva la presenza di due tipi di comandi:

1) i comandi  set  ed unset che permettono di assegnare al grafico un insieme di caratteristiche tali da determinarne il tipo e l'aspetto;

2) il comando  plot  che produce l'output del grafico sul terminale prescelto (nel nostro caso lo schermo in modalità  grafica).

In effetti l'unico comando di settaggio indispensabile è  unset parametric  e solo nel caso in cui durante la stessa sessione di  gnuplot  si sia precedentemente utilizzato  set parametric.

Gli stessi grafici si potevano ottenere col solo comando:

gnuplot>plot [0:2*pi] [-5:5] sin(x),cos(x),tan(x)
nel qual caso l'aspetto del grafico sarebbe stato quello di default. I valori chiusi tra parentesi quadre e separati dai due punti sono i valori minimo e massimo rispettivamente per  x  ed  y  (non obbligatori, infatti se omessi vengono definiti automaticamente dal programma).

Il comando  set samples 200  fissa a duecento il numero di punti nei quali viene calcolata la funzione (se omesso tale numero è per default 100). Aumentando il numero di punti aumenta la accuratezza con la quale la funzione sarà plottata, ma aumenta anche il tempo necessario a calcolarne i valori ed in ogni caso l'output sarà condizionato dalla definizione grafica del terminale d'uscita ( per grafica VGA standard è inutile superare 400 punti ed il più delle volte sono sufficienti 200 punti).

Come tracciare il grafico di una funzione del tipo  z = f(x,y)
Si tratta di una funzione non parametrica, ovvero di una funzione nella quale la variabile  z  è esplicitabile rispetto ad  x e ad  y  e per ciascuna coppia di valori di  x  e di  y  si ha un solo valore di  z. Tale funzione viene rappresentata da  gnuplot  con un grafico  3D  a superficie. Ad esempio si voglia tracciare il grafico di:

z = x2 – y2
gnuplot> unset parametric
gnuplot> set time
gnuplot> set title "es. grafico a superficie"
gnuplot> set xlabel "valori di x"

gnuplot> set ylabel "valori di y"

gnuplot> set zlabel "valori di z"
gnuplot> set samples 20,20
gnuplot> set isosamples 20,20
gnuplot> set xrange [-10:10]
gnuplot> set yrange [-10:10]
gnuplot> set zrange [-100:100]

gnuplot> f(x,y)=x**2-y**2
gnuplot> splot f(x,y)

Il significato dei vari comandi è facilmente interpretabile. Si osserva la presenza di set samples 20,20  che fissa a 20 il numero di punti equidistanti nei quali vengono campionati sia l'intervallo di definizione della variabile  x  sia l'intervallo di definizione della variabile  y . Questo significa che la funzione  z = f (x,y)  verrà calcolata per  400  coppie di valori  (x,y). Il comando set isosamples 20,20  fissa a  20  il numero di linee rispettivamente ad  x  e ad  y  costanti che determineranno il reticolo della rete con la quale si rappresenterà la superficie. Il comando che produce l'output del grafico 3D a superficie è  splot , gli intervalli per i valori di  x , y , z  sono opzionali.

Per la versione Windows del programma è disponibile il comando  set hidden3d  che permette di nascondere quelle parti di superficie che si trovano non in vista rispetto al punto di osservazione (tale comando non è utilizzabile per la versione MS-DOS).

Gli stessi risultati si potevano ottenere col solo comando:

gnuplot>splot [-10:10] [-10:10] [-100:100] x**2-y**2
Come tracciare il grafico di una funzione parametrica del tipo  f(x,y) = 0
Quando la funzione   f(x,y) = 0  è tale per cui risulta impossibile esplicitare la variabile dipendente  y  in quanto per lo stesso valore di  x  la  y  può assumere più valori (si pensi alla equazione della circonferenza), per tracciare il grafico si dovrà pensare ad una rappresentazione parametrica (razionale od in coordinate polari) della funzione.

La forma parametrica di una funzione è la seguente:   x = f1(t)  ,  y = f2(t)   dove  x , y sono le coppie  ascissa , ordinata  e  t  è il parametro. Vediamo un esmpio:

gnuplot> set parametric
gnuplot> set time
gnuplot> set grid
gnuplot> set title "spirale di Archimede"
gnuplot> set xlabel "ascissa"
gnuplot> set ylabel "ordinata"
gnuplot> set samples 300
gnuplot> set trange [0:10*pi]
gnuplot> set xrange [-35:35]
gnuplot> set yrange [-35:35]
gnuplot> plot t*cos(t),t*sin(t)
Il comando  set samples 300  fissa a  300  il numero di campionamenti equidistanti per l'intervallo del parametro  t   e quindi fissa a  300  il numero di coppie di valori  x , y  calcolati ed utilizzati per tracciare il grafico. Il comando plot richiede che siano fornite le due funzioni del parametro rappresentanti rispettivamente l’ascissa e l’ordinata.

Gli stessi risultati si potevano ottenere col solo comando:

gnuplot>plot [0:10*pi] [-35:35] [-35:35] t*cos(t),t*sin(t)
Come tracciare il grafico di una funzione parametrica del tipo  f(x,y,z) = 0
Il grafico 3D di una funzione    f(x,y,z) = 0    per la quale le coordinate siano espresse in forma parametrica del tipo   x = f1(u,v)  ,  y = f2(u,v)  ,  z = f3(u,v)  si può ottenere coi seguenti comandi:

gnuplot> set parametric
gnuplot> set time
gnuplot> set title "sfera"
gnuplot> set xlabel "asse x"
gnuplot> set ylabel "asse y"
gnuplot> set zlabel "asse z"
gnuplot> set samples 50
gnuplot> set isosamples 20,20
gnuplot> set urange [-pi:pi]
gnuplot> set vrange [-pi:pi]
gnuplot> set xrange [-2:2]
gnuplot> set yrange [-2:2]
gnuplot> set zrange [-0.7:0.7]
gnuplot> splot cos(u)*cos(v),cos(u)*sin(v),sin(u)
[image: image4.wmf]sfera

Fri Dec 24 15:05:02 1993

cos(u)*cos(v), cos(u)*sin(v), sin(u)

-2

-1.5

-1

-0.5

0

0.5

1

1.5

2

-2

-1.5

-1

-0.5

0

0.5

1

1.5

2

-1

-0.5

0

0.5

1

asse x

asse y

asse z


Vale la pena di evidenziare come ora i parametri siano due, rispettivamente  u  e  v e come il comando  set isosamples 20,20  definisca di riprodurre il reticolo della superficie tracciando  20  linee a  u  costante e  20  linee a  v  costante.

Come introdurre nelle espressioni le costanti complesse
Le costanti complesse possono essere espresse come {<real>,<imag>} dove <real> e <imag> devono essere costanti numeriche che rappresentano rispettivamente la parte reale e la parte immaginaria del numero complesso. Ad esempio {3,2} rappresenta il numero complesso 3+2i , {0,1} rappresenta l’unità immaginaria. Si ricorda che per ottenere i simboli { , } , bisogna attivare il tastierino numerico della tastiera (la spia NUM deve essere on) e, quindi, premere rispettivamente ALT123 ed ALT125, oppure usare i comandi EXPRESSIONS, COMPLEX NUMBER di gnuplot.
La possibilità di usare costanti complesse e l’unità immaginaria, unitamente alle funzioni abs, arg, imag, real, rende possibile disegnare i diagrammi di Bode e di Nyquist della risposta in frequenza di un dato sistema del quale sia nota la funzione di trasferimento. Ad esempio per il sistema avente la funzione di trasferimento:




Il diagramma di Bode del guadagno espresso in decibel per pulsazioni comprese tra 0,1 e 1000 [rad/s] risulta tracciato nel seguente modo (la pulsazione è indicata con la variabile x ):

gnuplot> unset parametric
gnuplot> set grid
gnuplot> set title "es. Bode guadagno"
gnuplot> set xlabel "pulsazione [rad/s]"
gnuplot> set ylabel "guadagno [db]"
gnuplot> set samples 1000
gnuplot> set xrange [0.1:1000]
gnuplot> set logscale x
gnuplot> plot 20*log10(abs(3/(5+2*{0,1}*x)))

[image: image5]
Il diagramma di Bode dell’argomento per pulsazioni comprese tra 0,1 e 1000 [rad/s] risulta tracciato nel seguente modo (la pulsazione è indicata con la variabile x ):
gnuplot> unset parametric
gnuplot> set grid
gnuplot> set title "es. Bode argomento"
gnuplot> set xlabel "pulsazione [rad/s]
gnuplot> set ylabel "argomento [°]"
gnuplot> set samples 1000
gnuplot> set xrange [0.1:1000]
gnuplot> set logscale x
gnuplot> set angles degrees
gnuplot> plot arg(3/(5+2*{0,1}*x))

[image: image6]
Il diagramma di Nyquist per pulsazioni comprese tra 0 e 1000 [rad/s] risulta tracciato come grafico parametrico nel seguente modo (la pulsazione è il parametro ed è indicata con la variabile t):

gnuplot> set parametric
gnuplot> set title "es. diagramma di Nyquist"
gnuplot> set xlabel "parte reale"
gnuplot> set ylabel "parte immeginaria"
gnuplot> set samples 10000
gnuplot> set trange [0:1000]
gnuplot> set grid
gnuplot> plot real((3/(5+2*{0,1}*t))),imag((3/(5+2*{0,1}*t)))

Come interpolare dei dati sperimentali mediante il comando fit e l'opzione smooth
Si sia condotta un'esperienza e si siano raccolti dei dati grazie ai quali a ciascun valore della grandezza x (ascissa) è associato un valore della grandezza y (ordinata) con un grado di attendibilità pari a p (peso, più esso è grande più il rilievo è attendibile, il peso è legato alla deviazione standard s). Si debba riportare su un sistema di assi coordinati le coppie (x,y) ovvero i punti rilevati sperimentalmente e quindi si debba interpolare i punti con una funzione f(x) la cui natura sia nota a priori. Allo scopo è disponibile il comando fit ed è necessario operare come segue:

1) salvare le terne x, y, s in un file dati del tipo testo al quale va data l'estensione .dat (allo scopo si può impiegare un qualsiasi editor testuale, ad esempio notepad.exe o edit.com). Nel file le righe che sono di commento devono cominciare col carattere # mentre i dati devono essere incolonnati così come appare nell'esempio sottostante (per meglio incolonnare si può usare il tasto di tabulazione):
# file egidio1.dat: contiene i dati per l’esempio d’uso del comando
# fit di gnuplot.
# ascissa   ordinata  standard deviation

0

  
0
          0.1

1.2
  
1
          1

2

  
4
          1

2.8
  
9.1            1

4.1
  
17
          1

5

  
23
          1

6

  
30
          1

7.5
  
55
          1

8

  
60
          1

9

  
85
          1

10

  
100            1

# fine file egidio1.dat

Il file “egidio1.dat” venga memorizzato nella sottodirectory  “miademo” della directory corrente. Quasi sempre la directory corrente è quella nella quale si trova il file eseguibile del programma gnuplot, ovvero C:\gnuplot\binaries\ e, in ogni caso, la directory corrente può essere individuata mediante il comando:
gnuplot> pwd
Si osserva che il terzo elemento di ogni terna deve essere la deviazione standard s che definisce il peso p da assegnare alla coppia  (x, y)  secondo la relazione:

Quindi se si vuole imporre il passaggio della funzione interpolante in prossimità di specifici punti basta assegnare a tali punti un valore basso, rispetto agli altri, di deviazione standard.

2) Sceglire la funzione f(x) con la quale si vuole interpolare i punti sperimentali, ad esempio:

f(x) = a·x2 + b·x + c  (quadratica, andamento parabolico)

3) Avviare gnuplot ed impartire i comandi che permettono di assegnare la funzione interpolante, calcolare i parametri a, b, c che ottimizzano la funzione interpolante rispetto ai punti assegnati, plottare sullo stesso grafico i punti sperimentali e la funzione interpolante:

gnuplot> unset parametric
gnuplot> set grid
gnuplot> set title "interpolazione fit"
gnuplot> set xlabel "ascissa"
gnuplot> set ylabel "ordinata"
gnuplot> f(x)=a*x**2+b*x+c
gnuplot> fit f(x) "./miademo/egidio1.dat" using 1:2:3 via a,b,c
gnuplot> plot "./miademo/egidio1.dat", f(x)
Il comando chiave è fit ed il suo significato è: interpola (fit) con la funzione f(x) i punti memorizzati nel file “egidio1.dat” utilizzando (using) per l’ascissa, l’ordinata e la deviazione standard  i valori nelle colonne 1, 2, 3 di tale file, tramite (via) il calcolo dei parametri a, b, c usati nella definizione della funzione f(x). In conclusione del comando fit, ai parametri a, b, c della funzione f(x) verranno assegnati definitivamente i valori calcolati. 
Si osservi come avviene la chiamata del file “egidio1.dat” da parte dei comandi di gnuplot (il file “egidio1.dat” è stato memorizzato nella sottodirectory  “miademo” della directory corrente).


[image: image9]
Se si desidera conoscere il valore che la funzione interpolante f(x) assume per un assegnato valore dell'ascissa x (ad esempio x = 4,3) basta dare il comando:

gnuplot> print f(4.3)
Naturalmente la scelta della funzione interpolante è arbitraria ed in base a tale scelta andrà assegnata la f(x) ed andranno assegnati i parametri a, b, c, … che la descrivono.

Nel caso non si desideri interpolare con un tipo specifico di funzione è disponibile l'opzione smooth del comando plot. Questa opzione rende disponibili cinque modi d'interpolazione: acsplines, bezier, csplines, sbezier, unique. Ancora una volta le terne x, y, p devono essere salvate in un file di testo analogo a quello già visto (solo che adesso il terzo dato per ogni punto deve essere esplicitamente il peso assegnato alla specifica misura). Ad esempio:

# file egidio2.dat: dati per l’esempio d’uso del comando plot smooth di

# gnuplot.
# ascissa    ordinata  peso 

0

    0

1

1.2

    1

1

2

    4

1

2.8

    9.1

1

4.1

    17

1

5

    23

1

6

    30

1

7.5

    55

1

8

    60

1

9

    85

1

10

   100

100

11

   90

1

12

   60

1

13

   50

1

14

   33

1

15

   24.5

1

16

   16

1

17

    9

1

18

    5

1

19

    2

1

20

    0

1

# fine file egidio2.dat

Quindi si eseguono i seguenti comandi di gnuplot:

gnuplot>unset parametric
gnuplot> set grid
gnuplot> set title "interpolazione acsplines"
gnuplot> set xlabel "ascissa"
gnuplot> set ylabel "ordinata"
gnuplot> plot "./miademo/egidio2.dat" using 1:2:3 smooth acsplines, "./miademo/egidio2.dat"
Il risultato è il seguente grafico:


[image: image10]
Quanto più si aumenta il peso p dei dati, tanto più la curva interpolante viene catturata dai punti assumendo così, a volte, un andamento maggiormente tortuoso.

